```
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %
%                                                                         %
%    File:           ch5_archi.dtr                                        %
%    Purpose:        Demonstrate defaults-based extended   deponency     %
%                                                                         %
%                                                                         %
%    Author:         Andrew Hippisley 23 06, 2010                         %
%    Email:          andrew.hippisley@uky.edu                             %
%    Address:        University of Kentucky, Lexington KY 40506-0027      %
%    Documentation:  Chapter 5, Morphological Mismatch & Extended         %
%                               Deponency                                 %
%    Related Files:  show_archi.dtr                                       %
%    Version:        1.02                                                 %
%                                                                         %
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %
%    Comment:  This theory accounts for Archi nouns: regular, deponent    %
%              and suppletive (section 5.4). Archi suppletive nouns       %
%              were discussed in chapter, page 16                         %
%                                                                         %
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %

% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %
%                                                                         %
%    1. SHOW AND HIDE                                                     %
%                                                                         %
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %

#load 'show_archi.dtr'.
#hide NOUN.

% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %
%                                                                         %
%    2. NOUN AS ROOT NODE                                                 %
%    Generalises the fact that morphological forms are based on the       %
%    various stems of  an Archi noun.                                     %
%                                                                         %
%    Also generalises the fact that the singular absolutive is built on   %
%    the root. For some nouns with two roots, it is the singular root     %
%    specifically                                                         %
%                                                                         %
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %


NOUN:
    <syn cat> == n
    <sem cat> == non_dynamic
    <syn> == "<mor>"
    <syn sg abs> == "<root sg abs>".


% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %
%                                                                         %
%    3. NOUN CLASSES                                                      %
%                                                                         %
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %


N_0:
    <sg> == "<root sg erg>"  N_0_SG:<>.
```

```
N_0_SG:
    <> == li MOR_CASE:<>
    <abs> == .

N_1:
    <> == N_0
    <pl> == "<root pl>" mul N_1_PL:<>.


N_1_PL:
    <> == čaj MOR_CASE:<>
    <abs> == .

N_2:
     <> == N_0
    <pl> ==  "<root pl>" N_2_PL:<>.


N_2_PL:
    <> == ttaj MOR_CASE:<>
    <abs> == ttu.

N_3:
    <sg> == "<root sg erg>" N_3_SG:<>
    <pl> == "<root pl>" N_3_PL:<>.


N_3_SG:
    <> == mu MOR_CASE:<>.


N_3_PL:
    <> == maj MOR_CASE:<>
    <abs> == .


% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %
%                                                                               %
%    4. CASE EXPONENTS                                                          %
%                                                                               %
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %


MOR_CASE:
    <erg> ==
    <gen> == n
    <dat> == s
    <comit> == ɬɬu
    <comp> == χur
    <perm> == kɬ’ana
    <part> == qˤiš
    <superlat> == ttik
    <sublat> == kɬ’ak.

% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %
%                                                                               %
%    5. LEXICAL ENTRIES: REGULAR                                                %
%    Each noun is associated with a stem formation class. Some nouns            %
%    have distinct stems for singular ergative forms and plural orms,           %
%    for example Kɬ’ánnu 'lover'.                                               %
```

```
%                                                                       %
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %


Arum:
   <> == NOUN
   <mor> == N_1:<>
   <gloss> == sickle
    <root> == aˤrum .

Aˤri:
    <> == NOUN
    <mor> == N_2:<>
    <gloss> == military division
    <root> == aˤri.

Kɬ’ánnu:
    <> == NOUN
    <mor> == N_3:<>
    <gloss> == lover
    <root> == kɬ’annu
    <root sg erg> == <root> m
    <root pl> == kɬ’annib.

% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %
%                                                                       %
%    6. LEXICAL ENTRIES: DEPONENT                                       %
%    Deponency involves number, specifically the productive stem-buildng%
%    process that associates a stem with a number, captured at the Stem %
%    Class nodes. The mismatch does not involve the root. In other words%
%    a plural deponent noun for plural morphosyntax uses an unexpected  %
%    singular stem but this is built on a plural root, the  same root   %
%    used for plural morphosyntax.                                      %
%                                                                       %
%    The deponency 'in the stems' may also involve heteroclisis, setting%
%    up another expectation that is not met: not only is the number     %
%    associated with the stem unexpected, the stem class referred to for%
%    building plural stem is different to the one used for building the %
%    singular stem.                                                     %
%                                                                       %
%    Finally, the deponency in Archi nouns is not accompanied by        %
%    defectiveness.                                                     %
%                                                                       %
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %

Haˤtəra:
   <> == NOUN
   <mor> == N_1:<>
   <mor sg> == <root> N_1_PL:<>
    <gloss> == river
    <root> == haˤtər
    <root sg abs> == haˤtəra.

C’aj:
   <> == NOUN
   <mor pl> == <root pl> N_1_PL:<>
   <mor sg> == <root> N_2_PL:<>
   <gloss> == goat
   <root> == c’aj
   <root pl> == c’ohor.
```

```
Xali:
     <> == NOUN
    <mor sg> == <root sg erg> N_3_PL:<>
    <mor> == N_2:<>
    <gloss> == family
    <root> == χali
    <root sg erg> == χal.

Xˤon:
    <> == NOUN
   <mor pl> == <root pl> N_0_SG:<>
   <mor sg> == <root sg erg> MOR_CASE:<>
   <gloss> == cow
    <root sg abs> == χˤon
    <root sg erg> == χˤini
    <root pl> == bucc'i.

% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %
%                                                                         %
%    7. LEXICAL ENTRIES: SUPPLETIVE                                       %
%    The two examples show suppletion involving parts of the paradigm     %
%    partitioned by the three stems: singular, ergative singular & plur.  %
%    These examples support the approach taken for regular deponent       %
%    nouns.  See chapter 1, examples (17) and (18)                        %
%                                                                         %
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %


Abttu:
    <> == NOUN
    <mor> == N_3:<>
    <gloss> == father
     <root sg abs> == abttu
    <root sg erg> == um.

Bič'ni:
    <> == NOUN
    <mor> == N_1:<>
    <mor pl> == <root pl> N_1_PL:<>
    <gloss> == corner of sack
    <root sg> == bič'ni
    <root pl> == boždo.
```